HRP 262 SAS LAB EIGHT, May 30, 2012


Lab Eight: Mixed models; time independent vs. time-dependent predictors

Lab Objectives

After today’s lab you should be able to:

1. Analyze longitudinal data with mixed models for repeated effects and mixed models with random effects.
2. Interpret results from (1).

3. Understand the difference between time-independent and time-dependent predictors.

4. Interpret results with time-independent predictors

5. Understand the difference between “between-subjects” and “within subjects” effects. 

6. Output predicted values from PROC MIXED and graph them.

SAS PROCs                           SAS EG equivalent

PROC MIXED

Analyze(ANOVA(Mixed models

PROC GPLOT

Graph ((Line Plot)
 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. For today’s class, download the lab 4-7 data at: www.stanford.edu/~kcobb/courses/hrp262. 
2. Open SAS EG; create a library pointing to the desktop.
3. Using code, turn the data into the long form, with both a continuous and categorical measure of time (time in months and dxa). Create both a repeated-measure outcome variable (bmc) and repeated-measure (=time-dependent) predictor (calcium). Do not fill in missing observations, since mixed models and GEE account for these.

data hrp262.runners;

set hrp262.runners;

id=_n_;

run;


data long;

set hrp262.runners;

dxa=1; time=0; bmc=bmc1; calc=calc1; output;

dxa=2; time=(dxaday2-dxaday1)*12/365.25; bmc=bmc2; calc=calc2; output;

dxa=3; time=(dxaday3-dxaday1)*12/365.25; bmc=bmc3; calc=calc3; output;

label time='Months since baseline';

label bmc='BMC (g)';

label calc='dietary calcium, mg/day';

run;

4. Recall GEE results from last time, for treatment group:

	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	2174.489
	43.3302
	2089.563
	2259.414
	50.18
	<.0001

	time
	0.8620
	0.4930
	-0.1042
	1.8282
	1.75
	0.0804

	treatr
	21.3704
	71.8467
	-119.447
	162.1874
	0.30
	0.7661

	time*treatr
	0.1248
	0.7609
	-1.3666
	1.6161
	0.16
	0.8698


For baseline calcium (time independent precictor):
	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	2179.000
	76.3872
	2029.284
	2328.716
	28.53
	<.0001

	calc1
	0.0017
	0.0431
	-0.0828
	0.0862
	0.04
	0.9687

	time
	-0.8399
	0.6728
	-2.1586
	0.4789
	-1.25
	0.2119

	calc1*time
	0.0012
	0.0004
	0.0004
	0.0020
	3.03
	0.0024


For calcium as a time-varying predictor:
	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	2185.019
	38.3723
	2109.810
	2260.227
	56.94
	<.0001

	calc
	-0.0053
	0.0136
	-0.0320
	0.0215
	-0.39
	0.6999

	time
	0.9131
	0.3956
	0.1376
	1.6885
	2.31
	0.0210


6. Now run a model in PROC MIXED using the repeated statement. This model directly models the within-subject correlation matrix, similar to GEE. It is basically the same approach as GEE—which is to deal with the intra-subject correlation by altering the covariance structure of the residuals. 
7. Run a MIXED model with treatment randomization as a predictor (time-independent predictor). Use the repeated effects option. Use the long form of the data. 

Analyze(ANOVA(Mixed models

Choose bmc as the dependent variable, treatr and time as the quantitative predictors, and 

Id as the classification variable. 

Under Fixed Effect Model, specify your model, including time and treatr as main effects and a time*treatr interaction. 

Under Fixed Effects Model Options, ask to see the parameter estimates! (Otherwise these will not be displayed!). Also, always use the KR option for estimating degrees of freedom:

In the Repeated Effects screen, choose ID as the subject identifier and compound symmetry for the covariance structure. This within-subject covariance matrix is called the R matrix. You can also ask to see this matrix. 

Then click Run. 

FYI, corresponding SAS code:
proc mixed data=long;

class  id; 

model bmc= time treatr time*treatr/solution ddfm=kr; 

repeated / subject=id type=cs;

run; quit;

	Estimated R Correlation Matrix for
id 1

	Row
	Col1
	Col2
	Col3

	1
	1.0000
	0.9806
	0.9806

	2
	0.9806
	1.0000
	0.9806

	3
	0.9806
	0.9806
	1.0000



	Fit Statistics

	-2 Res Log Likelihood
	2418.3

	AIC (smaller is better)
	2422.3

	AICC (smaller is better)
	2422.3

	BIC (smaller is better)
	2427.0




	Solution for Fixed Effects

	Effect
	Estimate
	Standard Error
	DF
	t Value
	Pr > |t|

	Intercept
	2174.46
	45.9232
	76.4
	47.35
	<.0001

	time
	0.8603
	0.3882
	119
	2.22
	0.0286

	treatr
	21.4722
	72.3736
	76.4
	0.30
	0.7675

	time*treatr
	0.1180
	0.6320
	119
	0.19
	0.8522


8. Use Modify Task to change the correlation structure to unstructured. What effect does it have on the AIC? 
9. An alternative approach to dealing with repeated measures is to leave the residuals alone, but actually alter the model by controlling for subject. Since we can’t fit a separate beta for every subject (as this would use up too many degrees of freedom), we instead can use random factors. Basically, we are allowing each subject to have their own intercept and potentially their own slope.   
What does it mean to have a random intercept? First look at the model with just a random intercept:

proc mixed data=long; 


class id;


model bmc = time  /s outpred=mixout ddfm=kr;


random int /subject=id;

run; quit;

goptions reset=symbol;

symbol1 i=join v=none l=1 r=12;

proc gplot data=mixout;

plot pred*time=id /nolegend;

run; quit;

10. What does it mean to have a random intercept and random time?
proc mixed data=long; 


class id;


model bmc = time  /s outpred=mixout ddfm=kr;


random int time/subject=id;

run; quit;

goptions reset=symbol;

symbol1 i=join v=none l=1 r=12;

proc gplot data=mixout;

plot pred*time=id /nolegend;

run; quit;

11. What does it mean to just have a random effect for time?
proc mixed data=long; 


class id;


model bmc = time  /s outpred=mixout ddfm=kr;


random time/subject=id;

run; quit;

goptions reset=symbol;

symbol1 i=join v=none l=1 r=12;

proc gplot data=mixout;

plot pred*time=id /nolegend;

run; quit;

12. Use code to run a mixed model with a random intercept. Note that you get identical results as with the use of the repeated statement. (PROC MIXED with repeated effects and compound symmetry = PROC MIXED with random intercept only). When you only include a random intercept in the model, the within-subject correlation is not allowed to change across different time points, so you are effectively forcing a compound symmetry structure for the within-subject correlation. 

proc mixed data=long;

class  id; 

model bmc= time treatr time*treatr/solution ddfm=kr; 

random int / subject=id ;

run; quit;

	Solution for Fixed Effects

	Effect
	Estimate
	Standard Error
	DF
	t Value
	Pr > |t|

	Intercept
	2174.46
	45.9232
	76.4
	47.35
	<.0001

	time
	0.8603
	0.3882
	119
	2.22
	0.0286

	treatr
	21.4722
	72.3736
	76.4
	0.30
	0.7675

	time*treatr
	0.1180
	0.6320
	119
	0.19
	0.8522


15. Run a mixed model with calc1 as a predictor (time-independent predictor). Start with a random intercept only. We have to do this with code:
proc mixed data=long;

class  id; 

model bmc= time calc1 time*calc1/solution ddfm=kr; 

random int / subject=id ;

run; quit;

	Fit Statistics

	-2 Res Log Likelihood
	2437.8

	AIC (smaller is better)
	2441.8

	AICC (smaller is better)
	2441.9

	BIC (smaller is better)
	2446.5


	Solution for Fixed Effects

	Effect
	Estimate
	Standard Error
	DF
	t Value
	Pr > |t|

	Intercept
	2179.02
	77.3810
	76.2
	28.16
	<.0001

	time
	-0.8590
	0.6377
	119
	-1.35
	0.1805

	calc1
	0.001691
	0.04747
	76.2
	0.04
	0.9717

	time*calc1
	0.001194
	0.000377
	119
	3.17
	0.0020


16. Now add a random slope; we can do this with point and click:
Analyze(ANOVA(Mixed Models

Choose bmc as the dependent variable, calc1 and time as the quantitative predictors, and 

Id as the classification variable. 

Under Fixed Effect Model, specify your model, including time and calc1 as main effects and a time*calc1 interaction. 

Under Fixed Effects Model Options, ask to see the parameter estimates! (Otherwise these will not be displayed!). Also, always use the KR option for estimating degrees of freedom:

Under Random Effects, click Add to add a random effect. Select time as the random effect; select include intercept as Yes; and select an unstructured covariance matrix. **This is the covariance matrix for the random effects (called the G matrix); unstructured is generally recommended here**


FYI, corresponding SAS code:

proc mixed data=long;

class  id; 

model bmc= time calc1 time*calc1/solution ddfm=kr;; 

random int time/ subject=id type=un ;

run; quit;

	Estimated G Matrix

	Row
	Effect
	id
	Col1
	Col2

	1
	time
	1
	2.6405
	41.8570

	2
	Intercept
	1
	41.8570
	95582


	Estimated G Correlation Matrix

	Row
	Effect
	id
	Col1
	Col2

	1
	time
	1
	1.0000
	0.08332

	2
	Intercept
	1
	0.08332
	1.0000


	Fit Statistics

	-2 Res Log Likelihood
	2432.9

	AIC (smaller is better)
	2440.9

	AICC (smaller is better)
	2441.1

	BIC (smaller is better)
	2450.3


	Solution for Fixed Effects

	Effect
	Estimate
	Standard Error
	DF
	t Value
	Pr > |t|

	Intercept
	2177.31
	76.6823
	74.9
	28.39
	<.0001

	time
	-0.7038
	0.7340
	60.2
	-0.96
	0.3414

	calc1
	0.003050
	0.04705
	74.9
	0.06
	0.9485

	time*calc1
	0.001065
	0.000443
	56.4
	2.40
	0.0196


More on time-changing predictors…

17. Graphing with two Y axes:
When you have time-changing predictors as well as time-changing outcomes, you may want to plot these simultaneously against time.
Graph the mean lines as follows (with a 1-standard deviation bar).

In the Long dataset, go to Graph > Line Plot. Select Multiple vertical column line plots using overlay. 

Under Data choose DXA (categorical time) as the Horizontal, bmc as the first Vertical, and calc as the second vertical.

Under Appearance > Plots, choose different line colors for the bmc and calc variables. Make sure you choose the same color for Data point marker (so the legend will have the correct colors)!

Under Interpolations choose STD as the Interpolation method. Check Compute the standard error of the mean and Join the means with a line. Last check the box next to Apply to all.

Click Run.

17. One way to look only at within-subject effects is to consider correlations between change scores:
data change;

set hrp262.runners;

ctime=(dxaday2-dxaday1)/12*365.25; cbmc=bmc2-bmc1; ccalc=calc2-calc1; output;

ctime=(dxaday3-dxaday2)/12*365.25; cbmc=bmc3-bmc2; ccalc=calc3-calc2; output;

run;

Plot and run regular linear regression on the change scores…

proc gplot data=change;

plot ccalc*ctime cbmc*ctime ccalc*cbmc;

symbol1 v=dot i=rl;

run;

proc reg data=change;

model cbmc= ctime ccalc;

run;





















Similar results to GEE! 














The treatr*time term represents the difference in the rate of change in bmc per month for the treatment vs. the control groups. 





The treatr term represents the baseline difference in bmc between the treatment and control groups. 





The time term represents the rate of change in bmc per month (.86 grams/month) for the control group. (significant)

Since the AIC improves very little, might not be worth making the model more complicated…




















You can see why we don’t need a random slope—there’s very little variation in the slopes between individuals.


























Note the assignment of permanent labels to variables within a datastep. This is a global assignment, so all subsequent plots and procedures will use these labels by default.























There is very little variability in the time slopes, indicating that you may not need a random slope in this model. 

















The intercept represents the average bmc at baseline for the control group. 





The AIC is 2422. We can compare this value to models in which we change the covariance structure.  








We need to assign a unique ID number to each participant for the long form (otherwise we will be unable to tell which observations belong to which subject)!








Calcium is a time-dependent predictor.

